

18 Solving Systems of Linear Differential Equations

This Chapter illustrates how Maple V's built-in linear algebra package can be useful in solving linear systems with constant coefficients.

We will use routines that come from the Maple V Library **linalg**. This library can be used in a session by invoking the command:

```
> with(linalg):
```

```
Warning: new definition for    norm
```

```
Warning: new definition for    trace
```

Note: In the above statement we used `:` instead of `;` to suppress output. The first example involves solving the homogeneous linear system

$$X' = AX,$$

where A is the 3 by 3 matrix:

$$A := \begin{bmatrix} -1 & 1 & 2 \\ 1 & 0 & 1 \\ 2 & 1 & -1 \end{bmatrix}$$

The way to solve such a system involves the following steps:

1. Find the eigenvalues and eigenvectors.
2. Write the general solution.

The first step can sometimes be very laborious to solve by hand even for systems as small as 3 by 3. Maple can be very useful in solving these problems. We now enter matrix.

```
> A := matrix(3,3,[-1,1,2,1,0,1,2,1,-1]);
```

$$A := \begin{bmatrix} -1 & 1 & 2 \\ 1 & 0 & 1 \\ 2 & 1 & -1 \end{bmatrix}$$

The command **eigenvects** gives the eigenvalues and corresponding eigenvectors in one step. If you enter the following command then Maple will return the eigenvalues, that except for their order should be as follows:

```
> eigsA := eigenvects(A);
```

$$\text{eigsA} := [-1, 1, \{[1 - 21]\}], [-3, 1, \{[-101]\}], [2, 1, \{[111]\}]$$

The last expression is a sequence of lists the first element of `eigsA`, `eigsA[1]`, is the list

$$[-3, 1, [-1, 0, 1]].$$

The first element of this list is given by `eigsA[1][1]` is -3 and is equal to one of the eigenvalues. The second element of this list, `eigsA[1][2]`, is the order of the eigenvalue (in this case 1). The final element in this list, `eigsA[1][3]`, is the set of eigenvectors corresponding to the eigenvalue, `eigsA[1][1]`. In this case there is only one eigenvector,

$$\text{eigsA}[1][3][1] = [-1, 0, 1].$$

Note: The order of the elements of `eigsA` vary from one time or another and thus sometime it might turn out that `eigsA[1]` is

$$[-1, 1, [1, -2, 1]].$$

In another Maple session the last Maple V command might lead to a different order. Once we have eigenvalues and enough independent eigenvectors to span the eigenspace of the A we can write the general solution.

```
> solA := sum(c[i]*exp(eigsA[i][1]*t)*eigsA[i][3][1], i=1..3);
```

$$solA := c_1 e^{(-t)} [1 - 2 \ 1] + c_2 e^{(-3t)} [-10 \ 1] + c_3 e^{(2t)} [1 \ 1 \ 1]$$

```
> solA := evalm(solA);
```

$$solA := \begin{bmatrix} c_1 e^{(-t)} - c_2 e^{(-3t)} + c_3 e^{(2t)} - 2 c_1 e^{(-t)} + c_3 e^{(2t)} c_1 e^{(-t)} + c_2 e^{(-3t)} + c_3 e^{(2t)} \\ \end{bmatrix}$$

The Maple command, **map**, allows to verify that `solA` is indeed a solution. Substituting $X = solA$ into $X' = A X$ first involves differentiating `solA`.

```
> map(diff, solA, t);
```

$$\begin{bmatrix} -c_1 e^{(-t)} + 3 c_2 e^{(-3t)} + 2 c_3 e^{(2t)} & 2 c_1 e^{(-t)} + 2 c_3 e^{(2t)} \\ -c_1 e^{(-t)} - 3 c_2 e^{(-3t)} + 2 c_3 e^{(2t)} \end{bmatrix}$$

The last result should equal $A X$.

```
> evalm(A &* solA);
```

$$\begin{bmatrix} -c_1 e^{(-t)} + 3 c_2 e^{(-3t)} + 2 c_3 e^{(2t)} & 2 c_1 e^{(-t)} + 2 c_3 e^{(2t)} \\ -c_1 e^{(-t)} - 3 c_2 e^{(-3t)} + 2 c_3 e^{(2t)} \end{bmatrix}$$

Thus `solA` is a solution. Since all of the eigenvalues are distinct we have an eigenspace with the right dimension and the solution is a general solution.

In applications we usually are required to solve initial value problems. The next thing we do is find the solution of $X' = A X$ that satisfies the initial conditions:

$$x_1(0) = 0, \quad x_2(0) = -2, \quad x_3(0) = 1.$$

First we solve for the constants c_1, c_2, c_3 .

```
> sol := simplify(solve({subs(t=0, solA[1])=0, subs(t=0, solA[2])=-2,
```

```
> subs(t=0, solA[3])=1}, {c[1], c[2], c[3]}));
```

$$\text{sol} := \left\{ c_2 = \frac{1}{2}, c_3 = \frac{-1}{3}, c_1 = \frac{5}{6} \right\}$$


We now substitute the values of c_1 , c_2 , and c_3 into the general solution to obtain the desired particular solution:

```
> solAa := [subs(sol,solA[1]),subs(sol,solA[2]),subs(sol,solA[3])];
```

$$\text{solAa} := \left[\frac{5}{6}e^{(-t)} - \frac{1}{2}e^{(-3t)} - \frac{1}{3}e^{(2t)}, -\frac{5}{3}e^{(-t)} - \frac{1}{3}e^{(2t)}, \frac{5}{6}e^{(-t)} + \frac{1}{2}e^{(-3t)} - \frac{1}{3}e^{(2t)} \right]$$


Once we know a particular solution we can ask for graphical information about the solution. For example a plot of x_1 versus time can be given by

```
> plot(solAa[1],t=-2..2,title = 'x1 vs t');
```



We can use a parametric plot to plot x_1 vs x_2 or x_1 vs x_3 or any other pair of dependent variables.


```
> plot([solAa[1],solAa[2],t=-2..2],title = 'x1 vs x2');
```



Since the solution curve lies in three dimensional space it is also useful to plot it using the Maple function **spacecurve** which is contained in the package **plots**. Thus we invoke **plots**.

```
> with(plots):
```

```
> spacecurve([solAa[1],solAa[2],solAa[3]],t=-2..2,axes = BOXED);
```



Of course **dsolve** can be used to find both general and particular solutions:

```
> eq1 := D(x1)(t) = A[1,1]* x1(t)+A[1,2]*x2(t)+A[1,3]*x3(t);
```

$$eq1 := D(x1)(t) = -x1(t) + x2(t) + 2x3(t)$$

```
> eq2 := D(x2)(t) = A[2,1]* x1(t) + A[2,2]*x2(t) + A[2,3]* x3(t);
```

$$eq2 := D(x2)(t) = x1(t) + x3(t)$$

```
> eq3 := D(x3)(t) = A[3,1]*x1(t) + A[3,2]*x2(t) + A[3,3]* x3(t);
```

$$eq3 := D(x3)(t) = 2x1(t) + x2(t) - x3(t)$$

```
> dsolve({eq1,eq2,eq3},{x1(t),x2(t),x3(t)});
```

$$\{x2(t) = _C1 e^{(2t)} - 2_C3 e^{(-t)}, x3(t) = _C1 e^{(2t)} + _C2 e^{(-3t)} + _C3 e^{(-t)}, \\ x1(t) = _C1 e^{(2t)} - _C2 e^{(-3t)} + _C3 e^{(-t)}\}$$

```
> dsolve({eq1,eq2,eq3,x1(0)=0,x2(0)=-2,x3(0)=1},{x1(t),x2(t),x3(t)});
```

$$\left\{ \begin{aligned} x2(t) &= -\frac{5}{3} e^{(-t)} - \frac{1}{3} e^{(2t)}, x1(t) = \frac{5}{6} e^{(-t)} - \frac{1}{2} e^{(-3t)} - \frac{1}{3} e^{(2t)}, \\ x3(t) &= \frac{5}{6} e^{(-t)} + \frac{1}{2} e^{(-3t)} - \frac{1}{3} e^{(2t)} \end{aligned} \right\}$$

The question that you should ask is: does this solution agree with the previous solution? Be careful, notice that the last solution is a set of equations and their order is unimportant.

If the eigenvalues are complex then the procedure is a little more complicated. The next example shows how to find the general solution of

$$X' = BX$$

where

$$B := \begin{bmatrix} 1 & 0 & 0 \\ 2 & 1 & -2 \\ 3 & 2 & 1 \end{bmatrix}$$

First enter this matrix. **Note:** There are two different techniques for entering a 3 by 3 matrix into a Maple session. Compare the command

```
> B := matrix([[1,0,0],[2,1,-2],[3,2,1]]);
```

we are about to use for entering B and the command

```
> A := matrix(3,3,[1,1,2,1,2,1,2,1,1]);
```

that we used for entering A. It seems that Maple treats both methods equally.

```
> B := matrix([[1,0,0],[2,1,-2],[3,2,1]]);
```

$$B := \begin{bmatrix} 1 & 0 & 0 \\ 2 & 1 & -2 \\ 3 & 2 & 1 \end{bmatrix}$$

As before, the solution of such a system requires that we

1. Find the eigenvalues and eigenvectors.
2. Write the general solution.

Again we use the Maple command **eigenvectors**.

```
> eigsB :=
> eigenvectors(B);
```

$$eigsB := \left[1, 1, \left\{ \begin{bmatrix} 1 & \frac{-3}{2} & 1 \end{bmatrix} \right\}, \right. \\ \left. \left[\text{RootOf}(-Z^2 - 2Z + 5), 1, \left\{ \begin{bmatrix} 0 & \frac{1}{2} \text{RootOf}(-Z^2 - 2Z + 5) - \frac{1}{2} & 1 \end{bmatrix} \right\} \right] \right]$$

The answer uses the notation

$$\text{RootOf}(-Z^2 - 2Z + 5)$$

for the complex eigenvalues in this case. We will show how to deal with this in a moment, but first we will write the solution corresponding to the real eigenvalue 1.

(Note: Remember that an item listed in a set is unordered. Hence the real eigenvalue may appear as `eigsB[j][1]` for `j=1, 2, or 3`. This particular time it occurs as the first element.)

```
> Sol[1] := exp(eigsB[1][1]*t)*eigsB[1][3][1];
```

$$Sol_1 := e^t \begin{bmatrix} 1 & \frac{-3}{2} & 1 \end{bmatrix}$$

```
> Sol[1] := evalm("");
```

$$Sol_1 := \left[e^t - \frac{3}{2} e^t e^t \right]$$

Turning to the complex eigenvalues, we will use the command **allvalues**.

```
> complexeig := allvalues(eigsB[2][1]);
```

$$complexeig := 1 + 2I, 1 - 2I$$

Substitute one component of the result of **allvalues** for the **RootOf** expression.

```
> comvect := subs(eigsB[2][1]=complexeig[1],eigsB[2][3]);
```

$$\text{comvect} := \{[0 \ 1 \ 1]\}$$

```
> comsol := evalm(exp(complexeig[1]*t)*comvect[1]);
```

$$\text{comsol} := [0 \ I e^{((1+2I)t)} e^{((1+2I)t)}]$$

Use **Re** and **Im** to obtain two real vector value solutions.

```
> Sol[2] := Re(comsol);
```

$$\text{Sol}_2 := \Re(\text{comsol})$$

```
> Sol[3] := Im(comsol);
```

$$\text{Sol}_3 := \Im(\text{comsol})$$

```
> evalm(Sol[2]);
```

$$[0 - \Im(e^{((1+2I)t)}) \Re(e^{((1+2I)t)})]$$

Construct the general solution and use **evalm** and **evalc** to evaluate the answer.

```
> solB := evalm(sum(c[i]*Sol[i], i=1..3));
```

$$\text{solB} := \begin{bmatrix} c_1 e^t - \frac{3}{2} c_1 e^t - c_2 \Im(e^{((1+2I)t)}) + c_3 \Re(e^{((1+2I)t)}) \\ c_1 e^t + c_2 \Re(e^{((1+2I)t)}) + c_3 \Im(e^{((1+2I)t)}) \end{bmatrix}$$

This does not look like a correct answer since Maple does not automatically simplify complex numbers. Thus we must apply the Maple command **evalc**.

```
> solB := map(evalc, solB);
```

$$\text{solB} := \begin{bmatrix} c_1 e^t - \frac{3}{2} c_1 e^t - c_2 e^t \sin(2t) + c_3 e^t \cos(2t) \\ c_1 e^t + c_2 e^t \cos(2t) + c_3 e^t \sin(2t) \end{bmatrix}$$

We can check to see if the equations are satisfied.

```
> map(diff, solB, t);
```

$$\begin{bmatrix} c_1 e^t - \frac{3}{2} c_1 e^t - c_2 e^t \sin(2t) - 2 c_2 e^t \cos(2t) + c_3 e^t \cos(2t) - 2 c_3 e^t \sin(2t) \\ c_1 e^t + c_2 e^t \cos(2t) - 2 c_2 e^t \sin(2t) + c_3 e^t \sin(2t) + 2 c_3 e^t \cos(2t) \end{bmatrix}$$

```
> evalm(B &* solB);
```

$$\begin{bmatrix} c_1 e^t - \frac{3}{2} c_1 e^t - c_2 e^t \sin(2t) - 2c_2 e^t \cos(2t) + c_3 e^t \cos(2t) - 2c_3 e^t \sin(2t) \\ c_1 e^t + c_2 e^t \cos(2t) - 2c_2 e^t \sin(2t) + c_3 e^t \sin(2t) + 2c_3 e^t \cos(2t) \end{bmatrix}$$

Let us now find a solution to an initial value problem and plot it.


```
> sol2 := simplify(solve({subs(t=0,solB[1])=
-1,subs(t=0,solB[2])=0,subs(t=0,solB[3])=1},{c[1],c[2],c[3]}));
```

$$sol2 := \left\{ c_1 = -1, c_2 = 2, c_3 = \frac{-3}{2} \right\}$$


```
> solBa := [subs(sol2,solB[1]),subs(sol2,solB[2]),subs(sol2,solB[3])];
```

$$solBa := \left[-e^t, \frac{3}{2} e^t - 2e^t \sin(2t) - \frac{3}{2} e^t \cos(2t), -e^t + 2e^t \cos(2t) - \frac{3}{2} e^t \sin(2t) \right]$$

```
> plot(solBa[2],t=-Pi..Pi,title = `x2 vs t`);
```




```
> plot([solBa[2],solBa[3],t=-Pi..Pi], title = `x2 vs x1`);
```



```
> with(plots):
```

```
> spacecurve([solBa[1],solBa[2],solBa[3]],t=-Pi..Pi,axes = NORMAL);
```



As before we can solve the differential equation directly using **dsolve**. First we enter the equation in the proper form.

```
> eqB1 := diff(x1(t), t) = B[1,1]*x1(t)+B[1,2]*x2(t) +B[1,3]*x3(t);
```

$$eqB1 := \frac{\partial}{\partial t} x1(t) = x1(t)$$

```
> eqB2 := diff(x2(t), t) = B[2,1]*x1(t)+B[2,2]*x2(t)+B[2,3]*x3(t);
```

$$eqB2 := \frac{\partial}{\partial t} x2(t) = 2x1(t) + x2(t) - 2x3(t)$$

```
> eqB3 := diff(x3(t), t) = B[3,1]*x1(t) + B[3,2]*x2(t)+B[3,3]*x3(t);
```

$$eqB3 := \frac{\partial}{\partial t} x3(t) = 3x1(t) + 2x2(t) + x3(t)$$

```
> dsolve({eqB1,eqB2,eqB3},{x1(t),x2(t),x3(t)});
```

$$\left\{ \begin{aligned} x3(t) &= -C1 e^t + C2 e^t \sin(2t) + C3 e^t \cos(2t), \\ x2(t) &= -\frac{3}{2} C1 e^t - C3 e^t \sin(2t) + C2 e^t \cos(2t), x1(t) = -C1 e^t \end{aligned} \right\}$$

```
> dsolve({eqB1,eqB2,eqB3,x1(0)=-1,x2(0)=0,x3(0)=1},{x1(t),x2(t),x3(t)});
```

$$\left\{ \begin{aligned} x1(t) &= -e^t, x3(t) = -e^t - \frac{3}{2} e^t \sin(2t) + 2 e^t \cos(2t), \\ x2(t) &= \frac{3}{2} e^t - 2 e^t \sin(2t) - \frac{3}{2} e^t \cos(2t) \end{aligned} \right\}$$

Exercises 18.0

1. Enter the matrix

$$A := \begin{bmatrix} 2 & 1 & 1 \\ 2 & 2 & -1 \\ -8 & -5 & -2 \end{bmatrix}$$

into a Maple V worksheet.

- (a) Use the Maple V command **eigenvectors** to find the eigenvalues and corresponding eigenvectors of A .
- (b) Write the general solution of the differential equation $X' = AX$.

2. Enter the matrix

$$B := \begin{bmatrix} -3 & 2 & 3 \\ 0 & -2 & 2 \\ 0 & -2 & -3 \end{bmatrix}$$

into a Maple V worksheet.

- (a) Use the Maple command **eigenvectors**, and **allvalues** to find the eigenvalues and corresponding real and complex eigenvectors of B .
- (b) Write the general solution of the differential equation $X' = BX$ in real form.