

15 Studying First Order Systems

In this lesson you will learn some tools for studying first order differential equations using Maple V. In particular, we will see how to find the general and particular solutions certain first order equations. In addition we will learn how to plot direction fields for first order scalar equations and systems of equations. Solution curves can also be added to these plots.

15.1 Analyzing First Order Scalar Equations

For our first example consider the first order differential equation

$$x' = x(2 - x).$$

```
> deq1 := diff(x(t), t) = x(t)*(2-x(t));
```

$$deq1 := \frac{\partial}{\partial t} x(t) = x(t)(-x(t) + 2)$$

Normally one uses the **dsolve** command to find the solution of a differential with Maple V, but, since this problem can be solved by separation of variables, we will now show how to use Maple V to solve it by that method.

```
> deqsep1 := deq1/(x(t)*(-x(t)+2));
```

$$deqsep1 := \frac{\frac{\partial}{\partial t} x(t)}{x(t)(-x(t) + 2)} = 1$$

The preceding command shows that the equation is of separable type. The next step is to integrate both sides. The next command integrates the left hand side of the equation using **lhs** and **int**.

```
> intlhsdeqsep1 := int(lhs(deqsep1), t);
```

$$intlhsdeqsep1 := \frac{1}{2} \ln(x(t)) - \frac{1}{2} \ln(x(t) - 2)$$

The right side is integrated and a constant of integration is added.

```
> intrhsdeqsep1 := int(rhs(deqsep1), t)+C;
```

$$intrhsdeqsep1 := t + C$$

Setting the two expression equal to one another and solving for $x(t)$ give us a solution which we denote by ϕ_1 .

```
> phi[1] := solve(intlhsdeqsep1=intrhsdeqsep1, x(t));
```

$$\phi_1 := -2 \frac{e^{(2t+2C)}}{1 - e^{(2t+2C)}}$$

The command **unapply** can be used to convert ϕ_1 from an expression to a function.

```
> phi[1] := unapply(phi[1], t);
```

$$\phi_1 := t \rightarrow -2 \frac{e^{(2t+2C)}}{1 - e^{(2t+2C)}}$$

It easy to check to see if ϕ_1 is a solution of the differential equation.

```
> eval(subs(x=phi[1],deq1));
```

$$-4 \frac{e^{(2t+2C)}}{1 - e^{(2t+2C)}} - 4 \frac{(e^{(2t+2C)})^2}{(1 - e^{(2t+2C)})^2} = -2 \frac{e^{(2t+2C)} \left(2 \frac{e^{(2t+2C)}}{1 - e^{(2t+2C)}} + 2 \right)}{1 - e^{(2t+2C)}}$$

```
> simplify(");
```

$$-4 \frac{e^{(2t+2C)}}{(-1 + e^{(2t+2C)})^2} = -4 \frac{e^{(2t+2C)}}{(-1 + e^{(2t+2C)})^2}$$

So substituting $x = \phi_1$ into the differential equation leads to the equality of both sides and hence ϕ is a solution, a general solution.

Next we show that the use of **dsolve** leads to the same solution. This is the most typical way to solve differential equations when using Maple V.

```
> dsolve(deq1,x(t));
```

$$\frac{1}{x(t)} = \frac{1}{2} + e^{(-2t)} _C1$$

In this result we have an “implicit” solution. Sometimes it is impossible even for Maple V to find the solution explicitly, but we can for this problem by using **solve**.

```
> phi[2] := solve(",x(t));
```

$$\phi_2 := -2 \frac{1}{-1 - 2e^{(-2t)} _C1}$$

Does this result agree with the previous one? The constants, C and $_C1$, are arbitrary. So substitute $_C1 = -\exp(-2 * C)/2$ into the latter solution.

```
> phi[2] := subs(_C1=-exp(-2*C)/2,phi[2]);
```

$$\phi_2 := -2 \frac{1}{-1 + e^{(-2t)} e^{(-2C)}}$$

The claim is that the function $\phi_1(t)$ and the expression ϕ_2 agree for all values of t . To see this consider the following:

```
> simplify(phi[2]-phi[1](t));
```

0

When we used **dsolve** above we obtained only an implicit solution, for which were able to solve for explicitly by using **solve**. The latter step could have been omitted had we used the option **explicit** in the **dsolve** command.

```
> dsolve(deq1,x(t),explicit);
```

$$x(t) = -2 \frac{1}{-1 - 2e^{(-2t)} - C1}$$

The **dsolve** command can be used to find the solution of an initial value problem. For example, suppose we wish to solve the following initial value problem.

$$x' = x(2 - x), \quad x(0) = 3.$$

Then we type in the following:

```
> dsolve({deq1,x(0)= 3},x(t),explicit);
```

$$x(t) = -2 \frac{1}{-1 + \frac{1}{3}e^{(-2t)}}$$

```
> phi[1] := rhs(");
```

$$\phi_1 := -2 \frac{1}{-1 + \frac{1}{3}e^{(-2t)}}$$

Next we find the solutions of

$$\begin{aligned} x' &= x(2 - x), & x(0) &= 3 \\ x' &= x(2 - x), & x(0) &= 0.1 \end{aligned}$$

and

$$x' = x(2 - x), \quad x(0) = -0.1.$$

```
> phi[2] := rhs(dsolve({deq1,x(0)= 1},x(t),explicit));
```

$$\phi_2 := -2 \frac{1}{-1 - e^{(-2t)}}$$

```
> phi[3] := rhs(dsolve({deq1,x(0)=0.1},x(t),explicit));
```

$$\phi_3 := -2 \frac{1}{-1 - 19.00000000e^{(-2t)}}$$


```
> phi[4] := rhs(dsolve({deq1,x(0)=-0.1},x(t),explicit));
```

$$\phi_4 := -2 \frac{1}{-1 + 21.00000000e^{(-2t)}}$$

A solution of an initial value problem is a parameterized curve which can be plotted. We plot all four curves on the same axes.

```
> P1 := plot({phi[1],phi[2] },t=0..2,x=0..3):
```

```
> P2 := plot({phi[3],phi[4]},t=0..2,x=-3..3):
> plots[display]({P1,P2});
```



Note that all of the curves that were initially positive seem to move toward the equilibrium solution $x(t) = 2$. Whereas the three curves that are initially smaller than 2 move away from the equilibrium solution $x(t) = 0$. This suggests that the first equilibrium solution is stable, and the second one is unstable.

Next we include the direction field of the given differential equation by using procedures contained in the **DEtools** package.


```
> with(DEtools):
```

The next plot uses the **DEtools** procedure **DEplot1** to produce a direction field plot.

```
> P3 := DEplot1(deq1,[t,x],t=0..2,x=-3..3):
```

This plot can be plotted along with the previous two plots to produce the following plot which contains the direction field along with four solution curves.

```
> plots[display]({P1,P2,P3},view = [0..2,-3..3]);
```



It is informative to study the linear differential equations resulting from linearizing the original equation near the equilibrium points $x = 0$ and $x = 2$. Recall that the linear approximation of a differentiable function $f(x)$ at a point a such that $f(a) = 0$ is

$$f(x) \approx f'(a)(x - a)$$

for x near a . Thus we obtain the derivative of the right hand side of the differential equation.

```
> fprime := D(x->x*(2-x));
```

$$fprime := x \rightarrow 2 - 2x$$

The equilibrium points are at $x = 0$ and $x = 2$ so we need $f'(0)$ and $f'(2)$.


```
> fprime(0);fprime(2);
```

2

-2


Recall that an equilibrium point $x = a$ is unstable if $f'(a) > 0$, and stable if $f'(a) < 0$. This confirms our guesses that $x = 0$ is unstable, and $x = 2$ is stable for our problem. We plot the direction field for the linearization near $x = 0$ along with two solution curves. This time we use **DEplot1** to plot the curves as well as the direction fields.

```
> P4 := DEplot1(fprime(0)*x, [t, x], t=0..2, x=-3..3):
> P5 := DEplot1(fprime(0)*x, [t, x], t=0..2, {x(0)=0.4, x(0)=-0.4},
> x=-3..3, arrows=NONE):
> plots[display]({P4, P5});
```



Observe that this plot has many of the same qualitative features that the original nonlinear differential equation possesses near the unstable equilibrium point $x = 0$. Now do the same thing about $x = 2$.

```
> P6 := DEplot1(fprime(2)*(x-2), [t, x], t=0..2, x=-3..3):
> P7 := DEplot1(fprime(2)*(x-2), [t, x], t=0..2, {x(0)=1, x(0)=3},
> x=-3..3, arrows=NONE):
> plots[display]({P6, P7});
```



Note that this plot shows that solutions of the linearization about $x = 2$ has nearly the same qualitative behavior around $x = 2$ as the original equation there.

Exercises 15.1 In the following problems we study the first order scalar differential equation

$$tx' = x \ln(tx) - x. \quad (1)$$

1. Use the Maple V command **dsolve** to find the general solution of the scalar differential equation (1) in explicit form. Your first Maple V statement should be something like the following.

```
> deq1 := t*diff(x(t),t) = x*ln(t*x(t))-x(t);
```

$$deq1 := t \frac{d}{dt} x(t) = x \ln(tx(t)) - x(t)$$

2. Find the solution of the initial value problem satisfying equation (??) and $x(1) = 2$.
3. Plot the integral curve found in part 2 for $1 \leq t \leq 2$, $0 \leq x \leq 2$.
4. Make a direction field plot for the differential equation (??), for $1 \leq t \leq 2$, $0 \leq x \leq 2$ along with the solution found in part 2.

15.2 Analyzing Systems of Equations

The next example shows how Maple V allows us to study a system of differential equations. We study the system:

$$\begin{aligned} x' &= xy \\ y' &= 2 - x(t). \end{aligned}$$

```
> deq2 := diff(x(t),t)=x(t)*y(t),diff(y(t),t)=-y(t)+2;
```


$$deq2 := \frac{\partial}{\partial t} x(t) = x(t)y(t), \frac{\partial}{\partial t} y(t) = -y(t) + 2$$

In this example we use the **DEtools** procedure **DEplot2**.

```
> P1 := DEplot2([deq2],[x,y],t=0..1,x=-2..2,y=-1..3):
```

The direction field for the system is given by the the following:

```
> plots[display](P1);
```




The next two commands produce a plot that creates some solution curves.

```
> P2 := DEplot2([deq2],[x,y],t=-1..2,{[x(0)=-1,y(0)=1],
[x(0)=-0.5,y(0)=1],[x(0)=0.5,y(0)=1],[x(0)=1,y(0)=1]},arrows=NONE):
```

```
> P3 := DEplot2([deq2],[x,y],t=-1..2,{[x(0)=0.4,y(0)=2.4],
[x(0)=-0.4,y(0)=2.4],[x(0)=-1,y(0)=3],[x(0)=1,y(0)=3]},arrows=NONE):
```

The next command plots the solution curves along with the direction field.

```
> plots[display]({P1,P2,P3},view=[-2..2,-1..3]);
```



Observe that this plot suggests that the point $(0, 2)$ is an equilibrium point. Let us now analyze the linearization of the system near this point. We need the **linalg** package.

```
> with(linalg):
```

```
Warning: new definition for norm
```

```
Warning: new definition for trace
```

The jacobian matrix evaluated at the point $(0, 2)$ represents the “derivative” of the right hand side

$$\langle xy, 2 - y \rangle$$

of the system.

```
> J := jacobian([x*y, 2-y], [x, y]);
```

$$J := \begin{bmatrix} y & x \\ 0 & -1 \end{bmatrix}$$

```
> J01 := subs(x=0, y=2, eval(J));
```

$$J01 := \begin{bmatrix} 2 & 0 \\ 0 & -1 \end{bmatrix}$$

The linearization of the system is given by multiplying the latter matrix by $\langle x, y - 2 \rangle$.

```
> A := multiply(J01, vector([x, y-2]));
```

$$A := [2x2 - y]$$


We now obtain solution curves and the direction field for the linearized system.

```
> P4 := DEplot2(convert(A, list), [x, y], t=0..1, x=-2..2, y=-1..3):
```

```
> P5 := DEplot2(convert(A, list), [x, y], t=-1..2, {[x(0)=-1, y(0)=1],
[x(0)=-0.5, y(0)=1], [x(0)=0.5, y(0)=1], [x(0)=1, y(0)=1]}, arrows=NONE):
```

```
> P6 := DEplot2([deq2], [x, y], t=-1..2, {[x(0)=0.4, y(0)=2.4],
[x(0)=-0.4, y(0)=2.4], [x(0)=-1, y(0)=3], [x(0)=1, y(0)=3]}, arrows=NONE):
```

```
> plots[display]({P4, P5, P6}, view=[-2..2, -1..3]);
```



Observe that the qualitative behavior of the phase plane of the linearization seems to agree with that of the non-linear system near the equilibrium.

We now consider another system of equations:

$$\begin{aligned}x' &= -x + xy/100, \\y' &= 2y - 2xy/25.\end{aligned}$$

```
> deq := diff(x(t),t) = -x(t)+x(t)*y(t)/100, diff(y(t),t) = 2*y(t)-2*x(t)*y(t)/25;
```

$$deq := \frac{d}{dt}x(t) = -x(t) + \frac{x(t)y(t)}{100}, \frac{d}{dt}y(t) = 2y(t) - \frac{2x(t)y(t)}{25}$$

Let us determine the equilibrium points for this system. The equilibrium points are those points where the right hand side of each equation vanishes. The **solve** command will do this.


```
> solve({rhs(deq[1])=0,rhs(deq[2])=0},{x(t),y(t)});
```

$$\{x(t) = 0, y(t) = 0\}, \{x(t) = 25, y(t) = 100\}$$

Thus we conclude that equilibrium points for this system are the points (0,0) and (25,100). Let us now plot the direction field for the system in the region x in $[-60,60]$, and y in $[-300,300]$.

```
> P1 := DEplot2([deq],[x,y],t=0..1,x=-60..60,y=-300..300):
```

```
> plots[display](P1);
```



There are a number of things to observe in the preceding plot. Note that the field becomes more interesting near the equilibrium points. Also note that the field is parallel to the coordinate axes at all points along the coordinate axes. This means that each coordinate axis is invariant relative to the flow of the system, in the sense that once a solution curve is on an axis it remains on that axis for all time. This also implies that each quadrant is invariant. Thus, for example, a solution curve which is initially in the first quadrant will remain in that quadrant. This is important in biological models where x and y may denote populations of certain species and negative populations make no sense. Let us now find the linear approximations to the system at the two equilibrium points. First we find the linearization at the point (0,0).

```
> J := jacobian([-x+x*y/100,2*y-2*x*y/25],[x,y]);
```

$$J := \begin{bmatrix} -1 + \frac{y}{100} & \frac{x}{100} \\ -\frac{2y}{25} & 2 - \frac{2x}{25} \end{bmatrix}$$


```
> J00 := subs(x=0,y=0,eval(J));
```

$$\begin{bmatrix} -1 & 0 \\ 0 & 2 \end{bmatrix}$$


This means that the linear approximation of the system near (0,0) is the vector field:

```
> V[1] := multiply(J00,[x,y]);
```

$$V_1 := [-x, 2y]$$

The direction field for this linearization is given by

```
> DEplot2(convert(V[1],list),[x,y],t=0..1,x=-20..20,y=-100..100);
```



Observe that, as in our previous examples, the qualitative behavior of the linear approximation of the system at the equilibrium point (0,0) is the same as that of the nonlinear system near (0,0). We now investigate the behavior of the linearization at the point (25,100).


```
> J25100 := subs(x=25,y=100,eval(J));
```

$$J_{25100} := \begin{bmatrix} 0 & 1/4 \\ -8 & 0 \end{bmatrix}$$

```
> V[2] := multiply(J25100,[x-25,y-100]);
```

$$V_2 := \left[\frac{y}{4} - 25, -8x + 200 \right]$$

```
> PL1 := DEplot2(convert(V[2],list),[x,y],t=0..1,x=5..40,y= 50..150):";
```




As we might now be able to predict the behavior of the linearized vector field is that of the original nonlinear system near the equilibrium point (25,100). For illustration purposes we will now plot this direction field with a couple of solution curves of the linear system.

```
> PL2 := DEplot2(convert(V[2],list),[x,y],t=0..10,{[x(0)=25,y(0)=60],
```


```
[x(0)=25,y(0)=75]},x=5..40,y= 50..150,stepsize=0.1,arrows=NONE):
```

```
> plots[display]({PL1,PL2});
```




In this case the solution curves form two concentric closed curves centered at the equilibrium point. Let us now compare the latter plot with solution curves with the same initial conditions for the original nonlinear system.

```
> P2 := DEplot2([deq],[x,y],t=0..10,{[x(0)=25,y(0)=60],[x(0)=25,y(0)=75]},
x=-60..60,y=-300..300,stepsize=0.1, arrows = NONE):
> plots[display]({P1,P2},view = [0..60,0..300]);
```




Notice that the solution curves of the nonlinear system near the equilibrium point (25,100) are like the solution curves for the linear approximation at that point. We will now plot time domain plots for the same two solutions of the nonlinear system. For a plot of x vs t we invoke the following command, where we use the **scene** option:

```
> DEplot2([deq],[x,y],t=0..10,{[x(0)=25,y(0)=60],[x(0)=25,y(0)=75]},
arrows=NONE, x=0..60,y=0..300,scene=[t,x],stepsize=0.1);
```



Similarly for y vs t .

```
> DEplot2([deq],[x,y],t=0..10,{[x(0)=25,y(0)=60],[x(0)=25,y(0)=75]},
arrows=NONE, x=0..60,y=0..300,scene=[t,y],stepsize=0.1);
```



Note that the curves in the last plot have larger amplitude than the curves in the x vs t plot. Also observe that both solutions seem to be periodic with period around 4.5.

Exercises 15.2 In the following we will study the first order system of differential equations

$$x' = 2x - 6xy/5, \quad (2)$$

$$y' = -y + 9xy/10. \quad (3)$$

Your first Maple V statement should be something like the following.

```
> deq2 := diff(x(t),t)=2*x(t)-6*x(t)*y(t)/5,
```

```
diff(y(t),t) = -y(t) + 9*x(t)*y(t)/10;
```

$$deq2 := \frac{d}{dt}x(t) = 2x(t) - \frac{6x(t)y(t)}{5}, \frac{d}{dt}y(t) = -y(t) + \frac{9x(t)y(t)}{10}$$

1. Plot the direction field for the system (??) - (??) for $0 \leq t \leq 1$, $0 \leq x \leq 5$, $0 \leq y \leq 5$.
2. Plot two solution curves of system (??)-(??) in the (x,y) plane that at time zero pass through the points (1,0.5) and (1,1), respectively, for $0 \leq t \leq 10$, $0 \leq x \leq 5$, $0 \leq y \leq 5$.
3. Plot a composite of the curves found in problem 2 and direction field found in problem 1.
4. Plot the two solutions from problem 2 in the (t,x)-plane, with $0 \leq t \leq 14$, $0 \leq x \leq 5$.
5. Plot the two solutions from problem 2 in the (t,y)-plane, with $0 \leq t \leq 14$, $0 \leq y \leq 5$.
6. Using the plots produced above estimate the period for each of the solutions.